/*
 * Licensed under the MIT License
 * 
 * Copyright (c) 2011 Kay Kreuning
 * 
 * Permission is hereby granted, free of charge, to any person obtaining a copy
 * of this software and associated documentation files (the "Software"), to deal
 * in the Software without restriction, including without limitation the rights
 * to use, copy, modify, merge, publish, distribute, sublicense, and/or sell
 * copies of the Software, and to permit persons to whom the Software is
 * furnished to do so, subject to the following conditions:
 * 
 * The above copyright notice and this permission notice shall be included in
 * all copies or substantial portions of the Software.
 * 
 * THE SOFTWARE IS PROVIDED "AS IS", WITHOUT WARRANTY OF ANY KIND, EXPRESS OR
 * IMPLIED, INCLUDING BUT NOT LIMITED TO THE WARRANTIES OF MERCHANTABILITY,
 * FITNESS FOR A PARTICULAR PURPOSE AND NONINFRINGEMENT. IN NO EVENT SHALL THE
 * AUTHORS OR COPYRIGHT HOLDERS BE LIABLE FOR ANY CLAIM, DAMAGES OR OTHER
 * LIABILITY, WHETHER IN AN ACTION OF CONTRACT, TORT OR OTHERWISE, ARISING FROM,
 * OUT OF OR IN CONNECTION WITH THE SOFTWARE OR THE USE OR OTHER DEALINGS IN
 * THE SOFTWARE.
 * 
 * http://code.google.com/p/hyves-as3-sdk/
 */
package nl.hyves.api.login
{
	import nl.hyves.api.methods.Auth;
	import nl.hyves.api.methods.options.ExpirationType;
	import nl.hyves.api.net.IHyvesResponse;

	import com.adobe.cairngorm.contract.Contract;

	import flash.display.NativeWindowInitOptions;
	import flash.events.Event;
	import flash.geom.Rectangle;
	import flash.html.HTMLLoader;
	import flash.net.URLRequest;

	public class AIRLogin extends AbstractLogin implements ILogin
	{
		public static var POPUP_DIMENSIONS:Rectangle = new Rectangle(100, 100, 500, 500);

		public function AIRLogin(methods:Vector.<String>, expirationType:ExpirationType = null)
		{
			expirationType ||= ExpirationType.DEFAULT;

			Contract.precondition(methods && methods.length > 0);
			Contract.precondition(expirationType != null);

			super(Auth.requesttoken(methods, expirationType), onRequesttoken);
		}

		override protected function onRequesttoken(response:IHyvesResponse):void
		{
			super.onRequesttoken(response);

			var options:NativeWindowInitOptions = new NativeWindowInitOptions();
			options.resizable = false;
			options.maximizable = false;

			var request:URLRequest = new URLRequest(buildAuthorizeURL(requesttoken.oauth_token));

			var loader:HTMLLoader = HTMLLoader.createRootWindow(true, options, false, POPUP_DIMENSIONS);
			loader.addEventListener(Event.LOCATION_CHANGE, onLocationChange);
			loader.load(request);
		}

		private function onLocationChange(event:Event):void
		{
			var loader:HTMLLoader = event.target as HTMLLoader;
			var location:String = decodeURIComponent(loader.location);

			if (/accepted/.test(location))
			{
				var token:String = location.match(/(?<=oauth_token=)[^&]+/g)[0];

				service.call(Auth.accesstoken(token), onAccesstoken, {api_immediate: true});

				loader.stage.nativeWindow.close();
			}
			else
			{
				// TODO: throw error
			}
		}
	}
}
